
ASIC	Application Specific Integrated Circuits; 

[bookmark: _Toc341963944]Formulácia úlohy
Navrhnite logický obvod na realizáciu operácie násobenia dvoch komplexných čísel. Reálna a imaginárna časť vstupných operandov je reprezentovaná v pevnej rádovej čiarke vo formáte : 4 bity pre vyjadrenie celej časti a 12 bitov na reprezentáciu desatinnej časti čísla. Na vyjadrenie celej časti výsledku výpočtu použite 8 bitov a 24 bitov pre desatinnú časť.
[bookmark: _Toc341963945]Analýza riešenia
Nech sú dané dve komplexné čísla  a . Ich súčin sa vyjadruje v tvare:
	(1)
Zo vzťahu (1) vyplýva, že na realizáciu výpočtu potrebujeme vykonať 4 násobenia, jedno odčítanie a jedno sčítanie nad číslami v pevnej rádovej čiarke. V prípade, že by sme sa rozhodli riešiť úlohu na báze predchádzajúcej úvahy náš obvod by zaberal veľa miesta na čipe. Kvôli optimalizácii veľkosti plochy čipu namiesto štyroch násobičiek, jednej odčítačky a jednej sčítačky použijeme jednu násobičku a jednu kombinovanú sčítačku-odčítačku. Prepojenie funkčných stupňov (násobička, sčítačka-odčítačka, pomocné registre, multiplexory) je znázornené na Obr. 1.

[image: D:\TUKE\Pedagogika\Predmety\Stavba pocitacov\Cvika\Ilustracny priklad fig1.jpg]
[bookmark: _Ref341955383][bookmark: _Toc341963958]Obr. 1	Prepojenie funkčných stupňov
[bookmark: _Toc341963946]Voľba návrhového prostriedku a typu realizačného obvodu
Návrhovým prostriedkom pre tvorbu funkčnej jednotky sa stal CAD systém od firmy XILINX s názvom ISE Design Suits 2012.3 v edícii ISE WebPACK [3]. Vzhľadom na skutočnosť, že operácie majú byť realizované v pevnej rádovej čiarke zvolila sa vývojová doska Spartan-6 SP605 Evaluation Platform (Obr. 2), ktorá natívne podporuje realizáciu výpočtov v pevnej rádovej čiarke [2]. Na syntézu obvodu sa použije nástroj XST [5], [7] a na simuláciu ISim [2], [4]. Opisným jazykom návrhu sa zvolil jazyk VHDL [5]  so zohľadnením štandardu VHDL-200X.
[image: I:\SPoc\design properties croped.jpg]
[bookmark: _Ref341955415][bookmark: _Toc341963959]Obr. 2	Nastavenia
[bookmark: _Ref101952784][bookmark: _Ref101960788][bookmark: _Toc102191183]
[bookmark: _Toc341963947]Návrh obvodu v jazyku VHDL
Na prácu s pevnou rádovou čiarkou sa použila knižnica IEEE_PROPOSED (Obr. 3). Použitie tejto knižnice je špecifické pre prostredie XILINX ISE Design Suits 2012.3 [2]. V inom prostredí (napr. Mentor HDL Designer 2012.1) sa použije balík IEEE.FIXED_PKG z knižnice IEEE.
---------------------------------------------------------------------------------------------
--  Deklarácia knižníc a balíkov pre prostredie XILINX ISE Design Suits 2012.3
library IEEE, IEEE_PROPOSED;
use IEEE.STD_LOGIC_1164.ALL;
use IEEE_PROPOSED.FIXED_FLOAT_TYPES.ALL;
use IEEE_PROPOSED.FIXED_PKG.ALL;
---------------------------------------------------------------------------------------------
--  Deklarácia knižníc a balíkov pre prostredie Mentor HDL Designer 2012.1
-- library IEEE;
-- use IEEE.STD_LOGIC_1164.ALL;
-- use IEEE.FIXED_PKG.ALL;
---------------------------------------------------------------------------------------------
[bookmark: _Ref341955440][bookmark: _Toc341963960]Obr. 3	Deklarácia knižníc a balíčkov
[bookmark: _Toc341963948]Definovanie vstupných a výstupných portov obvodu
Návrh začína špecifikáciou logického obvodu z pohľadu počtu a typu vstupných a výstupných portov (Tab. 1), ktorý je uvedený v deklaračnej časti bloku entity (Obr. 4).

[bookmark: _Ref341955461][bookmark: _Toc341963970]Tab. 1	Vstupné a výstupné porty obvodu
	Názov
	Smer
	Význam

	clk
	IN
	Hodinový signál, ktorý sa použije na synchronizáciu činnosti jednotlivých komponentov obvodu.

	reset
	IN
	Resetovací signál, ktorý sa použije na resetovanie pamäťovej časti obvodu.

	input_rdy
	IN
	Slúži na označenie dostupnosti operandov (a_r, b_r, a_i, b_i) na vstupe obvodu.

	a_r
	IN
	Reálna časť komplexného čísla A.

	a_i
	IN
	Imaginárna časť komplexného čísla A.

	b_r
	IN
	Reálna časť komplexného čísla B.

	b_i
	IN
	Imaginárna časť komplexného čísla B.

	p_r
	OUT
	Reálna časť výsledku operácie P = A*B.

	p_i
	OUT
	Imaginárna časť výsledku operácie P = A*B.



entity multiplier is
    Port ( clk : in  STD_LOGIC;
           reset : in  STD_LOGIC;
           input_rdy : in  STD_LOGIC;
           a_r, a_i : in  sfixed(3 downto -12);
           b_r, b_i : in  sfixed(3 downto -12);
           p_r, p_i : out sfixed(7 downto -24));
end multiplier;
[bookmark: _Ref341955479][bookmark: _Toc341963961]Obr. 4	Blok entity
[bookmark: _Toc341963949]Návrh operačnej časti obvodu
Navrhnutý obvod (Obr. 1) pozostáva z riadiacej (na obrázku vyznačená nie je) a z operačnej časti. Opis toku dát v jazyku VHDL operačnou časťou navrhnutého logického obvodu je znázornený na obrázku Obr. 5.
architecture rtl of multiplier is

  signal a_sel, b_sel,
         pp1_ce, pp2_ce,
         sub, p_r_ce, p_i_ce : std_logic;

  signal a_operand, b_operand : sfixed(3 downto -12);
  signal pp, pp1, pp2, sum	: sfixed(7 downto -24);
  signal plus, minus		: sfixed(8 downto -24);

  ...

begin

  a_operand <= a_r when a_sel = '0' else a_i;
  b_operand <= b_r when b_sel = '0' else b_i;

  pp <= a_operand * b_operand;

  pp1_reg : process (clk) is
  begin
    if rising_edge(clk) then
      if pp1_ce = '1' then
        pp1 <= pp;
      end if;
    end if;
  end process pp1_reg;
  
  pp2_reg : process (clk) is
  begin
    if rising_edge(clk) then
      if pp2_ce = '1' then
        pp2 <= pp;
      end if;
    end if;
  end process pp2_reg;


  plus <= pp1 + pp2;
  minus <= pp1 - pp2;
  sum <= plus(7 downto -24) when sub = '0' else minus(7 downto -24);

  p_r_reg : process (clk) is
  begin
    if rising_edge(clk) then
      if p_r_ce = '1' then
        p_r <= sum;
      end if;
    end if;
  end process p_r_reg;
  
  p_i_reg : process (clk) is
  begin
    if rising_edge(clk) then
      if p_i_ce = '1' then
        p_i <= sum;
      end if;
    end if;
  end process p_i_reg;

  ...

end architecture rtl;
[bookmark: _Ref341955910][bookmark: _Toc341963962]Obr. 5	Blok architecture pre reprezentáciu toku dát operačnou časťou obvodu

Signály uvedené v deklaračnej časti bloku architecture reprezentujú prepojenia medzi komponentmi obvodu a riadiace signály generované riadiacou časťou obvodu (Obr. 7). 
Multiplexory na vstupe obvodu (Obr. 1) sú reprezentované zápisom do signálov a_operand a b_operand na základe hodnoty riadiacich signálov a_sel a b_sel, ktoré fungujú ako selektory pre reálnu a imaginárnu časť vstupných komplexných čísel a a b.
Zápis do pomocného signálu pp rieši operáciu násobenia príslušných zložiek komplexných čísel podľa vzťahu (1). Procesy pp1_reg a pp2_reg fungujú ako záchytné registre pre čiastkové súčiny
 	(2)

Výstup z kombinovanej sčítačky-odčítačky je prezentovaný signálom sum. 
[bookmark: _Toc341963950]Návrh riadiacej časti obvodu
Návrh riadiacej časti obvodu je založený na určení postupnosti krokov na realizáciu výpočtu podľa vzťahu (1). Existuje niekoľko možností v akom poradí je možné vykonať kroky výpočtu. V nasledujúcej časti dokumentu sa uvedie jedna z možností. 
[bookmark: _Toc341963951]Kroky algoritmu výpočtu
Kvôli zjednodušeniu návrhu, povolí sa aktivácia v danom hodinovom takte (clk) len jedného funkčného stupňa obvodu. Výpočet je odštartovaný nastavením signálu input_rdy = ‘1’.

	Krok.1
	Vynásob operandy a_r a b_r. Ulož výsledok súčinu do registra 1.

	Krok.2
	Vynásob operandy a_i a b_i. Ulož výsledok súčinu do registra 2.

	Krok.3
	Odčítaj od hodnoty v registri 1 hodnotu v registri 2. Ulož výsledok výpočtu do registra pre uchovanie reálnej časti výsledku P.

	Krok.4
	Vynásob operandy a_r a b_i. Ulož výsledok súčinu do registra 1.

	Krok.5
	Vynásob operandy a_i a b_r. Ulož výsledok súčinu do registra 2.

	Krok.6
	Pripočítaj k hodnote z registra 1 hodnotu z registra 2. Ulož výsledok výpočtu do registra pre uchovanie imaginárnej časti výsledku P.



Riadiaci automat (RA) sa navrhne na báze automatu Moore. Zlúčením krokov 3 a 4 (medzi krokmi 3 a 4 sa neprejaví zdrojový hazard), počet stavov RA sa rovná číslu päť. Počet potrebných riadiacich signálov, ich názvy ako aj informácia o tom, v ktorom hodinovom takte (clk), ktorý signál má byť aktívny sa uvádza v Tab. 2.

[bookmark: _Ref341959019][bookmark: _Toc341963971]Tab. 2	Riadiace signály
	stav/krok
	a_sel
	b_sel
	pp1_ce
	pp2_ce
	sub
	p_r_ce
	p_i_ce

	1
	0
	0
	1
	0
	--
	0
	0

	2
	1
	1
	0
	1
	--
	0
	0

	3
	0
	1
	1
	0
	1
	1
	0

	4
	1
	0
	0
	1
	--
	0
	0

	5
	--
	--
	0
	0
	0
	0
	1



Potom stavový graf (Obr. 6) RA typu Moore má tvar
[image: I:\SPoc\f4-34-P369528.jpg]
[bookmark: _Ref341959435][bookmark: _Toc341963963]Obr. 6	Stavový graf riadiaceho automatu

Opis riadiacej časti obvodu je znázornený na Obr. 7.

architecture rtl of multiplier is

  signal a_sel, b_sel,
         pp1_ce, pp2_ce,
         sub, p_r_ce, p_i_ce : std_logic;

  signal a_operand, b_operand : sfixed(3 downto -12);
  signal pp, pp1, pp2, sum    : sfixed(7 downto -24);
  signal plus, minus    		: sfixed(8 downto -24);

  type multiplier_state is (step1, step2, step3, step4, step5);
  signal current_state, next_state : multiplier_state;

begin

  ...

  state_reg : process (clk, reset) is
  begin
    if reset = '1' then
      current_state <= step1;
    elsif rising_edge(clk) then
      current_state <= next_state;
    end if;
  end process state_reg;
  
  next_state_logic : process (current_state, input_rdy) is
  begin
    case current_state is
      when step1 =>
        if input_rdy = '0' then
          next_state <= step1;
        else
          next_state <= step2;
        end if;
      when step2 =>
        next_state <= step3;
      when step3 =>
        next_state <= step4;
      when step4 =>
        next_state <= step5;
      when step5 =>
        next_state <= step1;
    end case;
  end process next_state_logic;

   output_logic : process (current_state) is
   begin
     a_sel <= '0'; b_sel <= '0';  pp1_ce <= '0'; pp2_ce <= '0';
     sub <= '0';   p_r_ce <= '0'; p_i_ce <= '0';
     case current_state is
       when step1 =>
         pp1_ce <= '1';
       when step2 =>
         a_sel <= '1'; b_sel <= '1';  pp2_ce <= '1';
       when step3 =>
         b_sel <= '1'; pp1_ce <= '1'; sub <= '1'; p_r_ce <= '1';
       when step4 =>
         a_sel <= '1'; pp2_ce <= '1';
       when step5 =>
         p_i_ce <= '1';
     end case;
   end process output_logic;

end architecture rtl;
[bookmark: _Ref341963012][bookmark: _Toc341963964]Obr. 7	Blok architecture pre reprezentáciu toku dát riadiacou časťou obvodu

[bookmark: _Toc341963952]Syntéza obvodu
Nasledujúci výpis generovaný počas RTL syntézy obvodu nám poskytuje informáciu o kódovaní stavov stavového automatu a o časovaní signálov.
=============================================================
*	Low Level Synthesis 	*
=============================================================
Analyzing FSM <MFsm> for best encoding.
Optimizing FSM <FSM_0> on signal <current_state[1:5]> with one-hot encoding.
-------------------
State | Encoding
-------------------
 step1 | 00001
 step2 | 00010
 step3 | 00100
 step4 | 01000
 step5 | 10000
-------------------
Timing Summary:
---------------
Speed Grade: -3

   Minimum period: 6.468ns (Maximum Frequency: 154.598MHz)
   Minimum input arrival time before clock: 6.984ns
   Maximum output required time after clock: 3.597ns
   Maximum combinational path delay: No path found
[bookmark: _Toc341963965]Obr. 8	Výpis generovaný počas RTL syntézy nástrojom XST

Využiteľnosť komponentov dosky je znázornená v Tab. 3. 

[bookmark: _Ref341959798][bookmark: _Toc341963972]Tab. 3	Využiteľnosť komponentov vývojovej dosky Spartan-6 SP605 Evaluation Platform
	Device Utilization Summary (estimated values)
	[-]

	Logic Utilization
	Used
	Available
	Utilization

	Number of Slice Registers
	135
	54576
	0%

	Number of Slice LUTs
	132
	27288
	0%

	Number of fully used LUT-FF pairs
	51
	216
	23%

	Number of bonded IOBs
	131
	296
	44%

	Number of BUFG/BUFGCTRLs
	1
	16
	6%

	Number of DSP48A1s
	1
	58
	1%



RTL schéma zapojenia obvodu je znázornená na Obr. 9.
[image: I:\SPoc\rtl synteza croped.jpg]
[bookmark: _Ref341963430][bookmark: _Toc341963966]Obr. 9	RTL schéme obvodu
[bookmark: _Toc341963953]Simulácia činnosti obvodu
Na simuláciu činnosti obvodu sa použil nástroj ISim, ktorý však nie je schopný znázorniť čísla vo formáte pevnej rádovej čiarky (sfixed), preto sa použila hexadecimálna reprezentácia pre operandy a_r, b_r, a_i, b_i, p_r a p_i (Obr. 12).
Na overenie korektnosti návrhu sa použila simulačná zostava pozostávajúca zo stimulačného procesu (Obr. 10) 
  ...

  reset <= '1', '0' after 2 * clk_period;

  apply_test_cases : process is

    procedure apply_test ( a_test, b_test : in complex ) is
    begin
      a <= a_test; b <= b_test; input_rdy <= '1';
      wait until falling_edge(clk); input_rdy <= '0';
      for i in 1 to 5 loop
        wait until falling_edge(clk);
      end loop;
    end procedure apply_test;

  begin
    wait until falling_edge(clk) and reset = '0';
    apply_test(cmplx(0.0, 0.0), cmplx(1.0, 2.0));
    apply_test(cmplx(1.0, 1.0), cmplx(1.0, 1.0));
    -- further test cases ...
    wait;
  end process apply_test_cases;

  a_r <= to_sfixed(a.re, a_r'left, a_r'right);
  a_i <= to_sfixed(a.im, a_i'left, a_i'right);
  b_r <= to_sfixed(b.re, b_r'left, b_r'right);
  b_i <= to_sfixed(b.im, b_i'left, b_i'right);

  ...
[bookmark: _Ref341960343][bookmark: _Toc341963967]Obr. 10	Stimulačný proces
a z overovacieho procesu (Obr. 11).
  ...

  check_outputs : process is
    variable p : complex;
  begin
    wait until rising_edge(clk) and input_rdy = '1';
    p := a * b;
    for i in 1 to 5 loop
      wait until falling_edge(clk);
    end loop;
    assert abs (to_real(p_r) - p.re) < 2.0**(-12) and abs (to_real(p_i) - p.im) < 2.0**(-12);
  end process check_outputs;  

  ...
[bookmark: _Ref341960865][bookmark: _Toc341963968]Obr. 11	Overovací proces

Na základe vykonanej simulácie, simulačné priebehy (Obr. 12) dokazujú, že obvod pracuje správne.

[image: I:\SPoc\ise simulacia.JPG]
[bookmark: _Ref341960023][bookmark: _Toc341963969]Obr. 12	Simulačný výstup
[bookmark: _Toc341963954]Návrh na urýchlenie výpočtu 
V prípade, že sa použijú dve násobičky, výpočet reálnej a imaginárnej časti výsledku násobenia dvoch komplexných čísel môže byť realizovaný v rovnakom hodinovom takte (clk). Tento predpoklad nám umožňuje zrealizovať výpočet za tri kroky :
	Krok.1
	Vynásob a_r a b_r a ulož výsledok do registra 1; vynásob a_i a b_i a ulož výsledok do registra 2;

	Krok.2
	Odčítaj od obsahu regista 1 obsah registra 2 a ulož výsledok výpočtu do registra pre reálnu časť výsledku P; vynásob a_r a b_i a ulož výsledok do registra 1; vynásob a_i a b_r a ulož výsledok do registra 2;

	Krok.3
	Pripočítaj k obsahu regista 1 obsah registra 2 a ulož výsledok výpočtu do registra pre imaginárnu časť výsledku P;





[bookmark: _Toc102191192][bookmark: _Toc224306322][bookmark: _Toc341963955]Záver
Úlohou tejto práce bolo navrhnúť logický obvod na realizáciu operácie násobenia dvoch komplexných čísel v pevnej rádovej čiarke. Navrhnutý obvod dokáže pracovať na frekvencií 154.598MHz. ....

